MATHEMATICA の経済学への応用 (3)

―― 生産関数，利潤，および Averch=Johnson モデル ――

鵜 沢 秀[1)]

0．はじめに

いくつかの投入物（生産要素）を組み合わせて，生産物を効率的に生産するプロセス（技術）を表現する一つの方法として生産関数によるアプローチがある。生産関数は理論分析や実証分析を問わず，生産活動や経済活動を解明するときに用いられる，重要な手段となっている。

技術を表現するには，任意の大きさの生産物を生産するのに最小の費用で生産する方法を示す，費用関数による表現もある。

生産関数と費用関数との双対性に関する議論を含む，理論的な研究の成果は数多くなされており，Shephard［1953，1970］，Diewert［1982］，および Nadiri［1982］などにまとめられている。

また，計量経済分析を中心にした実証研究では，後に，コブ＝ダグラス生産関数と呼ばれる関数を示した，Cobb and Douglas［1928］や CES 生産関数を提言した，Arrow，Chenery，Minhas，and Solow［1961］などの研究成果が利用されてきている。これらの成果を集計理論に応用した，Sato［1975］の仕事は有名である。さらに，Christensen, Jorgenson and Lau［1971，1973］などによる，transcendental logarithmic[2)]生産関数や費用関数を利用した実証研究は非常に多くなっている。

1）小樽商科大学商学部経済学科

2）略して，trans-log（トランスログ）とも言う。

他方，数式処理ソフトの MATHEMATICA を利用して[3)]，コブ＝ダグラス生産関数の特徴を図形的に明らかにする仕事は Huang and Crooke［1997］，Stinespring［2002］，および小林［1996］などによって実行されている。この種の研究は経済学教育の補助に主眼をおいたアプローチである。

代替の弾力性が一定（Constant Elasticity of Substitution，略して CES という）の生産関数はパラメータの特別な値に対して，コブ＝ダクラス生産関数，レオンティエフ型生産関数，および完全代替型の生産関数を含んでいる。しかも非常に操作しやすい生産関数であることが数学的な検討でこれまで知られている。MATHEMATICA の数式処理と描画能力を用いたアニメーション画像の表示により，CES 生産関数の持つ特徴をこの小論で明らかにする。この分野では初めての試みである。

次に，コブ＝ダグラス生産関数を利用し，独占企業の利潤最大化行動を分析する。最初に，目的関数が収入の最大化と利潤の最大化では，（目的関数の値を最大化するという意味で）最適な労働投入量と資本投入量の組み合わせが異なることを確認する。また，Averch and Johnson［1962］によって理論的に検討された，公正報酬率規制がない場合と公正報酬率規制がある場合の均衡値の違いを MATHEMATICA の数式処理とアニメーション作画能力を用いて，比較検討する。

この小論の構成は以下のようになっている。第1節では， MATHEMATICA を利用し，コブ＝ダクラス生産関数を含む，CES 生産関数の2つのパラメータを操作することにより，その図形的特徴を明らかにする。第2節では，コブ＝ダグラス生産関数の持つ，図形的特徴を簡単に紹介する。第3節と第4節では，右下がりの需要曲線に直面する，独占企業の利潤最大化問題を考察する。第3節では，3次元空間における収入曲面と（費用平面に平行な）補助平面を利用して，最大利潤を探す。第4節では，3次元空間における利潤曲面と（利

3）MATHEMATICA を利用した経済学への応用例として，Varian［1993，1996］，浅利・久保・石橋・山下［1997］，鵜沢［1996，1998，2000a，2000b，2002，2005，2006，2007］などがある。

潤水準を示す）補助平面を利用して，最大利潤を探す。第５節では，コブ＝ダグラス生産関数と費用関数を利用し，アバーチ＝ジョンソン・モデル（公正報酬率規制のない場合と公正報酬率規制を課された場合とでは均衡での労働投入量と資本投入量の組（L,K）はどのような影響を受けるか）を MATHEMATICA の数式処理と描画能力を用いて解明する。最後に結語が述べられる。

１．CES 生産関数

１．１　代替の弾力性が一定の生産関数

２つの生産要素（たとえば，労働と資本）を用いて，一つの生産物を効率的に生産する技術を持つ企業を考える。代替の弾力性とは生産要素価格の比（賃金率／レンタル料）が１パーセント変化したときに，同じ生産量を生産するために必要な生産要素の大きさの比（労働投入量／資本投入量）が何パーセント変化するかを示す概念である。

いま，代替の弾力性が一定（Constant Elasticity of Substitution）の CES 生産関数[4)]を考えよう。CES 生産関数は $\mathrm{Y}=\mathrm{A}\,\{\alpha \mathrm{L}^{-\rho}+(1-\alpha)\mathrm{K}^{-\rho}\}^{-\varepsilon/\rho}$ であらわされる。ここで，Y は生産量，L は労働投入量，K は資本投入量をあらわし，A は効率性に関するパラメータ，α は分配に関するパラメータ，ρ は代替の弾力性に関するパラメータ[5)]，および ε は規模に関する収穫（returns to scale）のパラメータ[6)]を示している。

MATHEMATICA では，規模に関して収穫一定の CES 生産関数は次のように表記することができる[7)]。

４）代替の弾力性および CES 生産関数の特徴については，早見・鵜沢・若林・今・佐竹［1986，pp.54-59］，西村［1990，pp.193-200］などを参照せよ。

５）後に示すように，代替の弾力性 σ は（$1+\rho$）の逆数に等しい。

６）以下の議論では簡単化のため，規模に関する収穫一定の生産関数，すなわち，$\varepsilon=1$ の場合に限定して検討する。

７）In[n]:＝は MATHEMATICA が自動的につける入力記号である。同様に，Out[n]＝は出力記号である。以下の例が示すように，n には番号を示す数字が入っている。

```
In[1]:= cesY[L_, K_] := A (α L^(-ρ) + (1 - α) K^(-ρ))^(-1/ρ)
```

ここで，cesYはCES生産関数の場合の生産量，Lは労働投入量，Kは資本投入量をあらわし，Aは効率性に関するパラメータ，αは分配に関するパラメータ，およびρは代替の弾力性に関するパラメータをあらわす。

最初に，生産関数を労働について偏微分することにより，労働の限界生産物，mpL，を求めよう。

```
In[2]:= mpL = D[cesY[L, K], L]
```

Out[2]= $A L^{-1-\rho} \alpha (K^{-\rho} (1-\alpha) + L^{-\rho} \alpha)^{-1-\frac{1}{\rho}}$

同様に，生産関数を資本について偏微分することにより，資本の限界生産物，mpK，を求める。

```
In[3]:= mpK = D[cesY[L, K], K]
```

Out[3]= $A K^{-1-\rho} (1-\alpha) (K^{-\rho} (1-\alpha) + L^{-\rho} \alpha)^{-1-\frac{1}{\rho}}$

労働の限界生産物が資本の増加によりどのように変化するかを示す，交差偏微分，mpLK，を求めておこう。

```
In[4]:= mpLK = D[mpL, K]
```

Out[4]= $-A K^{-1-\rho} L^{-1-\rho} (1-\alpha) \alpha (K^{-\rho} (1-\alpha) + L^{-\rho} \alpha)^{-2-\frac{1}{\rho}} \left(-1-\frac{1}{\rho}\right) \rho$

次に，生産関数を労働について2回偏微分する。結果を簡単化するために，MATHEMATICAではSimplifyを用いる。

```
In[5]:= mpLL = Simplify[D[mpL, L]]
```

$$\text{Out[5]=}\quad \frac{A K^{\rho} L^{-2+\rho} (-1+\alpha)\,\alpha\,(-K^{-\rho}(-1+\alpha)+L^{-\rho}\alpha)^{-1/\rho}(1+\rho)}{(L^{\rho}(-1+\alpha)-K^{\rho}\alpha)^{2}}$$

同様に，生産関数を資本について 2 回偏微分する。

```
In[6]:= mpKK = Simplify[D[mpK, K]]
```

$$\text{Out[6]=}\quad \frac{A K^{-2+\rho} L^{\rho} (-1+\alpha)\,\alpha\,(-K^{-\rho}(-1+\alpha)+L^{-\rho}\alpha)^{-1/\rho}(1+\rho)}{(L^{\rho}(-1+\alpha)-K^{\rho}\alpha)^{2}}$$

さて，代替の弾力性 σ を次に求める。その定義より[8)]，

```
In[7]:= σ = mpL mpK (mpL L + mpK K) / (L K (2 mpLK mpL mpK - mpLL mpK^2 - mpKK mpL^2))
```

$$\text{Out[7]=}\quad \Bigg(A^{2} K^{-2-\rho} L^{-2-\rho}(1-\alpha)\,\alpha\,(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-2-\frac{2}{\rho}}$$
$$\left(A K^{-\rho}(1-\alpha)(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-1-\frac{1}{\rho}}+A L^{-\rho}\alpha(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-1-\frac{1}{\rho}}\right)\Bigg)\Bigg/$$
$$\Bigg(-2A^{3}K^{-2-2\rho}L^{-2-2\rho}(1-\alpha)^{2}\alpha^{2}(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-4-\frac{3}{\rho}}\left(-1-\frac{1}{\rho}\right)\rho-$$
$$\frac{1}{(L^{\rho}(-1+\alpha)-K^{\rho}\alpha)^{2}}\Big(A^{3}K^{-2-\rho}L^{-2+\rho}(1-\alpha)^{2}(-1+\alpha)\,\alpha$$
$$(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-2-\frac{2}{\rho}}(-K^{-\rho}(-1+\alpha)+L^{-\rho}\alpha)^{-1/\rho}(1+\rho)\Big)-$$
$$\frac{A^{3}K^{-2+\rho}L^{-2-\rho}(-1+\alpha)\,\alpha^{3}(K^{-\rho}(1-\alpha)+L^{-\rho}\alpha)^{-2-\frac{2}{\rho}}(-K^{-\rho}(-1+\alpha)+L^{-\rho}\alpha)^{-1/\rho}(1+\rho)}{(L^{\rho}(-1+\alpha)-K^{\rho}\alpha)^{2}}\Bigg)$$

上の結果を簡単化すると代替の弾力性は（$1+\rho$）の逆数に等しいことがわかる。

8）西村［1990，p.196］の式（6.65）を用いる。いま，生産関数を $y=f(x_1, x_2)$ とする。ここで，y は生産量，x_1，x_2 はそれぞれ別の生産要素とする。$f_i=\partial f(x_1, x_2)/\partial x_i$（$i=1, 2$），$f_{ij}=\partial^2 f(x_1, x_2)/\partial x_i \partial x_j$（$i, j=1, 2$）とおくと，この生産関数の場合の代替の弾力性 σ は $\sigma=f_1 f_2(f_1 x_1+f_2 x_2)/\{x_1 x_2(2f_{12}f_1 f_2-f_{11}f_2^{2}-f_{22}f_1^{2})\}$ で与えられる。

```
In[8]:= Simplify[σ]
```

$$\text{Out[8]=}\ \frac{1}{1+\rho}$$

次のように，代替の弾力性を求める計算式の分子と分母を別々に計算してから σ を求めることもできる。

```
In[9]:= bunshi = Simplify[mpL mpK (mpL L + mpK K)]
```

$$\text{Out[9]=}\ -\frac{A^3 K^{-1+\rho} L^{-1+\rho} (-1+\alpha)\,\alpha\,(-K^{-\rho}(-1+\alpha) + L^{-\rho}\alpha)^{-3/\rho}}{(L^\rho(-1+\alpha) - K^\rho \alpha)^2}$$

```
In[10]:= bunbo = Simplify[L K (2 mpLK mpL mpK - mpLL mpK^2 - mpKK mpL^2)]
```

$$\text{Out[10]=}\ -\frac{A^3 K^{-1+\rho} L^{-1+\rho} (-1+\alpha)\,\alpha\,(-K^{-\rho}(-1+\alpha) + L^{-\rho}\alpha)^{-3/\rho}(1+\rho)}{(L^\rho(-1+\alpha) - K^\rho \alpha)^2}$$

```
In[11]:= σ = bunshi / bunbo
```

$$\text{Out[11]=}\ \frac{1}{1+\rho}$$

1．2　多様な生産関数を CES 生産関数は表現できる[9)]

CES 生産関数はパラメータ ρ の値により，いろいろなタイプの生産関数を表現できることが知られている[10)]。

(i)　$\rho=0$ のときはコブ＝ダグクラス型生産関数となる。

このことを MATHEMATICA では関数 **Limit** を用いて確かめることができる。

9）代替の弾力性が一定という制約があることに注意せよ。より一般的な生産関数の例としては，第1節に述べたような transcendental logarithmic 生産関数などがあり，それらは代替の弾力性が可変である。Christensen, Jorgenson, and Lau［1971，1973］や Nadiri［1982］などを参照せよ。

10）例えば，西村［1990，pp.197-200］を参照せよ。

```
In[12]:= Limit[cesY[L, K], ρ → 0]
```

$$\text{Out[12]=}\ A\,K^{1-\alpha}\,L^{\alpha}$$

上の式は確かに，コブ＝ダグラス生産関数[11)]を示している。

(ii) $\rho = -1$ のときは完全代替の生産関数となる。

このことを MATHEMATICA で確かめよう。

```
In[13]:= Limit[cesY[L, K], ρ → -1]
```

$$\text{Out[13]=}\ A\,(K - K\,\alpha + L\,\alpha)$$

上の式は確かに，完全代替の生産関数を示している[12)]。

(iii) $\rho = \infty$ のときは完全非代替の生産関数，いわゆる，レオンティエフ型生産関数となる[13)]。

しかしながら，Stinespring［2002，p.122］も述べているように，MATHEMATICA は $\rho = \infty$ のときレオンティエフ型生産関数を見つけることができない。MATHEMATICA では次のように表示されるだけである。

```
In[14]:= Limit[cesY[L, K], ρ → Infinity]
```

$$\text{Out[14]=}\ \text{Limit}[A\,(K^{-\rho}\,(1-\alpha) + L^{-\rho}\,\alpha)^{-1/\rho},\ \rho \to \infty]$$

十分大きな ρ（例えば，21000）のときの生産関数のグラフ（これを生産曲

11) すなわち，生産量を Y とすると，$Y = AL^{\alpha}K^{1-\alpha}$ となっている。
12) すなわち，生産量を Y とすると，$Y = A\{\alpha L + (1-\alpha)K\}$ となっている。
13) 生産量を Y とし，式で示すと，Y＝A Min[L,K] となる。ある生産量をもたらす，効率的な投入量の組み合わせ（L,K）の集合を L−K 平面にあらわしたものを等量曲線とすると，この場合の等量曲線は45度線上で直角に折れ曲がった，L 字形となる。例えば，西村［1990，pp.198-199］を参照せよ。より一般的なレオンティエフ型生産関数は Y＝A Min[aL,bK] であらわされる。ここで，a と b は非負のパラメータである。

面と呼ぶことにする）を見てみると，これはレオンティエフ型の生産関数のグラフにほぼ等しいことがわかる。しかも，図1から図6を検討すると，ρ が無限大になると，α の値に依存しないことが予想できる。すなわち，生産量をYとし，式で示すと，Y＝A Min[L,K] となることが見て取れる。



図1　生産曲面
$\alpha=0.1$，$\rho=21000$



図2　生産曲面
$\alpha=0.5$，$\rho=21000$



図3　生産曲面
$\alpha=0.8$，$\rho=21000$



図4　図1の等量曲線図
$\alpha=0.1$，$\rho=21000$



図5　図2の等量曲線図
$\alpha=0.5$，$\rho=21000$



図6　図3の等量曲線図
$\alpha=0.8$，$\rho=21000$

図1から図3までの色のグラデーションは生産量水準の範囲に依存した，色分けがなされている。色分けは10クラスに分割されている。

また，等量曲線図を示す図4から図6は，図1から図3をそれぞれ，ほぼ真上から見た図として作成されたものである。ある等量曲線は，与えられた生産量を生産するのに必要な，効率的な労働投入量と資本投入量の組（L,K）の集合を示している。特に，色分けの境界で示されている等量曲線がほぼ45度線上で直角に折れ曲がった，L字形をしていることが読み取れる。

1．3　CES 生産関数のグラフ

パラメータ α と ρ の値に対し，CES 生産関数のグラフを次に描こう[14)]。ρ の値が－1のとき，完全代替の生産関数のグラフが描かれている。ρ の値がゼロに近いときは，コブ＝ダグラス生産関数のグラフに近似していることが読み取れる。また，既に指摘したように，ρ の値が100を超えて大きくなると，レオンティエフ型生産関数のグラフに近似することも確認できる。



図7　$\alpha=0.1$，$\rho=-1$



図8　$\alpha=0.5$，$\rho=-1$



図9　$\alpha=0.8$，$\rho=-1$



図10　$\alpha=0.1$，$\rho=-0.4$



図11　$\alpha=0.5$，$\rho=-0.4$



図12　$\alpha=0.8$，$\rho=-0.4$

14）紙幅の制約のため，$\alpha=0.1$, 0.5, 0.8および $\rho=-1$，-0.4，-0.2，-0.01, 1.1, 12, 102の場合を掲載した。また，白黒印刷のためのカラー配色をしている。自然なグラデーションを施した画像を白黒印刷すると，印刷結果がぼやけてしまうためである。1．4　白黒印刷でカラーグラデーションを表現する場合の困難性を見よ。



図13　$\alpha=0.1$, $\rho=-0.2$



図14　$\alpha=0.5$, $\rho=-0.2$



図15　$\alpha=0.8$, $\rho=-0.2$



図16　$\alpha=0.1$, $\rho=-0.01$



図17　$\alpha=0.5$, $\rho=-0.01$



図18　$\alpha=0.8$, $\rho=-0.01$



図19　$\alpha=0.1$, $\rho=1.1$



図20　$\alpha=0.5$, $\rho=1.1$



図21　$\alpha=0.8$, $\rho=1.1$



図22　$\alpha=0.1$, $\rho=2$



図23　$\alpha=0.5$, $\rho=2$



図24　$\alpha=0.8$, $\rho=2$



図25　$\alpha=0.1$，$\rho=12$



図26　$\alpha=0.5$，$\rho=12$



図27　$\alpha=0.8$，$\rho=12$



図28　$\alpha=0.1$，$\rho=102$



図29　$\alpha=0.5$，$\rho=102$



図30　$\alpha=0.8$，$\rho=102$

図7から図30までに描かれた，生産曲面の一覧図を縦に見ると，同じαの値に対してρが異なる場合の特徴を読み取ることができる。また，生産曲面の一覧表を横に見ると，同じρの値に対して，αの値が異なる場合の特徴を読み取ることが可能となる[15)]。

1．4　白黒印刷でカラー・グラデーションを表現する場合の困難性

同じαとρの値に対する生産曲面をカラー・グラデーションの異なる，4つの画像を見てみよう。まったく同じ内容を表示していても，見る人により印象が大きく変わるであろう。

---

15) 私の Web サイトにこの様子をカラー・グラデーションのアニメーションで表示する予定である。URL：http://www.otaru-uc.ac.jp/~uzawa/fj-papers.html よりたどれるように準備中である。



図31　$\alpha=0.4$，$\rho=0.2$



図32　$\alpha=0.4$，$\rho=0.2$



図33　$\alpha=0.4$，$\rho=0.2$



図34　$\alpha=0.4$,$\rho=0.2$

2．コブ＝ダグラス生産関数

2．1　生産曲面の形状はαに依存する

以下の分析では，一次同次のコブ＝ダグラス生産関数を考える。一次同次のコブ＝ダグラス生産関数をMATHEMATICAで表現すると[16)]，

16）以下では簡単化のために，MATHEMATICAが自動的につける入力記号，In[n] :=と出力記号，Out[n]=を省略した。

```
Y[L_,K_]:=A L^α K^(1-α)
```

となる。ここで，Y は生産量，L は労働投入量，K は資本投入量を示す。また，A は生産の効率を示すパラメータ，α は労働分配率，および $1-\alpha$ は資本分配率をそれぞれ示すパラメータである。α は 0 以上 1 以下の数である。

α の大きさにより生産曲面（生産関数を 3 次元のグラフに表示したもの）は多様な形状を示すことを確認できる。以下の図35より図37を参照されたい[17)]。また，参考のために，同じ生産量をもたらす労働投入量と資本投入量の組（L,K）の集合を描いた等量曲線図（図38より図40）をそれぞれの利潤曲面の下に掲載した。



図35　生産曲面
$\alpha=0.05$



図36　生産曲面
$\alpha=0.5$



図37　生産曲面
$\alpha=0.75$



図38　図35の等量曲線図
$\alpha=0.05$



図39　図36の等量曲線図
$\alpha=0.5$



図40　図37の等量曲線図
$\alpha=0.75$

17）作成した画像は21枚からなるアニメーションである。ここではその一部を掲載した。

２．２　収入を最大にする労働投入量と資本投入量の組（L,K）を求める

消費者の行動を示す，需要曲線（逆需要関数）を MATHEMATICA で，

```
p=a-b Y[L,K]
```

$$a - A\,b\,K^{1-\alpha}\,L^{\alpha}$$

とあらわす。ここで，p は価格を示し，a および b は正数のパラメータである。

収入 revenue は価格 p×販売量 Y[L,K] に等しいので，

```
revenue=p Y[L,K]
```

$$A\,K^{1-\alpha}\,L^{\alpha}\,(a - A\,b\,K^{1-\alpha}\,L^{\alpha})$$

となる。

収入 revenue を生産量 Y で表現すると，

```
revenue =(a-b Y) Y
```

となる。収入を最大にする生産量[18]を求めるには，収入 revenue を生産量 Y について微分し，微係数をゼロとする値として求めることができる。

```
sol=Solve[D[revenue,Y]==0,Y]
```

$$\left\{\left\{Y \to \frac{a}{2\,b}\right\}\right\}$$

```
revMaxY=Y/.sol
```

$$\left\{\frac{a}{2\,b}\right\}$$

となる。すなわち，Y＝a/(2b) で収入は最大となる。したがって，収入を最大にする労働投入量と資本投入量の組（L,K）は等生産量曲線

18）収入関数は生産量について２次関数であるので，最大化のための十分条件を満たしている。

```
Y[L,K]==revMaxY
```

$$A\,K^{1-\alpha}\,L^{\alpha} == \left\{\frac{a}{2\,b}\right\}$$

上にある。このことを，画像で確かめてみよう。

いま，A＝1，α＝0.6，a＝50，b＝1の値を前提に，収入曲面を描く[19]。



図41　収入曲面



図42　収入曲面



図43　収入曲面



図44　収入曲面



図45　等収入曲線

図44および図45は収入曲面の一部とそれを等収入曲線の図にしたものである。等収入曲線とは，同じ収入をもたらす，労働投入量と資本投入量の組（L,K）

19）作成した画像は24枚からなるアニメーションである。ここではその一部を掲載した。

の集合を示す。黒色と灰色との境界は収入水準500に対応し，灰色と白色との境界は収入水準600に対応する。真ん中の点線模様の線が最大の収入水準625に対応している（図54も参照せよ）。

収入を最大化する投入量の組と利潤最大をもたらす投入量の組とが異なることを確認するために，最初に，グラフを用いて収入最大点を見つけよう。そのために，収入曲面をL－K平面に平行な補助平面（収入の大きさをあらわし，画像では黒色で表示されている）で切断した画像を描いてみる。この補助平面の上に収入曲面の一部分が見えている限り，収入は最大化されていないことに注意しよう。



図46　収入水準＝0



図47　収入水準＝200



図48　収入水準＝400



図49　収入水準＝600



図50　収入水準＝622



図51　収入水準＝623



図52　収入水準＝624



図53　収入水準＝625

図46から図52において，切断する補助平面の上に収入曲面の一部分が見えているので，収入水準が624を達成するまでは労働投入量と資本投入量の組(L,K)の集合が読み取れ，収入は最大化されていない。

収入水準625（生産量が25＝50/2で，価格が25＝50－25である）のときは，労働投入量と資本投入量の組（L,K）の集合のグラフは切断する補助平面と重なってしまったために，見えなくなっていることに注意しよう（図53)。このとき，収入は最大になっている。図54には，収入を最大にする生産量（Y＝25)をもたらす労働投入量と資本投入量の組（L,K）の集合を示している。



図54　収入最大をもたらす労働投入量と資本投入量の組（L,K）の集合

2．3　利潤最大をもたらす労働投入量と資本投入量の組（L,K）を求める

コブ＝ダグラス生産関数を持つ独占企業が市場需要曲線に直面するとき，利潤最大をもたらす労働投入量と資本投入量の組（L,K）をMATHEMATICAの数式処理を利用して求めよう。

次のような費用関数を想定する。すなわち，労働の賃金率wを，資本のレンタル料をrとし，費用関数をcostとして，

```
cost=w L + r K
K r+L w
```

とする。

利潤をpaiであらわすと，利潤＝収入－費用であるので，

```
pai=(a-b Y)  Y - cost
-K r-L w+Y (a-b Y)
```

収入関数Rは，

```
R=p Y[L,K]
```
$$A K^{1-\alpha} L^{\alpha} (a - A b K^{1-\alpha} L^{\alpha})$$

とあらわすこともできる。

利潤極大化のための必要条件は，

① 労働の限界生産物収入が賃金率 w に等しい

② 資本の限界生産物収入がレンタル料 r に等しい

ことである[20)]。したがって，「2つの生産要素に関する限界生産物収入の比が要素価格比に等しい」ことを利用し，資本投入量と労働投入量の関係を導く。

労働に関する限界生産物収入の大きさ，mrL，は R を L に関して偏微分すると得られるので，

```
D[R,L]
```
$$-A^2 b K^{2-2\alpha} L^{-1+2\alpha} \alpha + A K^{1-\alpha} L^{-1+\alpha} (a - A b K^{1-\alpha} L^{\alpha}) \alpha$$
```
mrL=Simplify[D[R,L]]
```
$$A K^{1-2\alpha} L^{-1+\alpha} (a K^{\alpha} - 2 A b K L^{\alpha}) \alpha$$

同様に，資本に関する限界生産物収入の大きさ，mrK，は R を K に関して偏微分すると得られる。

```
D[R,K]
```
$$-A^2 b K^{1-2\alpha} L^{2\alpha} (1-\alpha) + A K^{-\alpha} L^{\alpha} (a - A b K^{1-\alpha} L^{\alpha}) (1-\alpha)$$
```
mrK=Simplify[D[R,K]]
```
$$A K^{-2\alpha} L^{\alpha} (-a K^{\alpha} + 2 A b K L^{\alpha}) (-1+\alpha)$$

上の2つの式を利潤極大のための必要条件に代入すると，資本投入量 K と労働投入量 L の間には以下の関係が成立する。

---

20) 西村［1990，p.216］を参照せよ。完全競争の場合は限界収入が価格に等しいので，ある生産要素の限界生産物収入は，その生産要素の価値限界生産物に等しくなる。

```
solK=Flatten[Simplify[Solve[mrL/mrK==w/r,K]]]
```

$$\left\{K \to -\frac{L\,w\,(-1+\alpha)}{r\,\alpha}\right\}$$

資本－労働比率（K/L）を KoverL であらわすと，

```
KoverL=K/L/.solK
```

$$-\frac{w\,(-1+\alpha)}{r\,\alpha}$$

となる。

資本の限界生産物収入， mrK，は $(1-\alpha)A(a-2bY)L^{\alpha}K^{-\alpha}$ であるが，$Y=K(L^{\alpha}K^{-\alpha})$，$L^{\alpha}K^{-\alpha}=(L/K)^{\alpha}=(K/L)^{-\alpha}=1/(K/L)^{\alpha}$ となることを考慮すると，

```
mrK=(a-2 b A K (1/KoverL)^α) A (1-α) (1/KoverL)^α
```

$$A\,(1-\alpha)\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}\left(a-2\,A\,b\,K\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}\right)$$

したがって，mrK=rとなる条件より，利潤最大をもたらす資本投入量，KKopt，を求めることができる。

```
solKK=Flatten[Solve[(a-2 b A K (1/KoverL)^α) A (1-α)
(1/KoverL)^α ==r,K]]
```

$$\left\{K \to \frac{\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{-2\,\alpha}\left(r-a\,A\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}+a\,A\,\alpha\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}\right)}{2\,A^{2}\,b\,(-1+\alpha)}\right\}$$

```
solKK=Flatten[Solve[mrK==r,K]]
```

$$\left\{K \to \frac{\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{-2\,\alpha}\left(r-a\,A\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}+a\,A\,\alpha\left(-\frac{r\,\alpha}{w\,(-1+\alpha)}\right)^{\alpha}\right)}{2\,A^{2}\,b\,(-1+\alpha)}\right\}$$

```
KKopt=Simplify[K/.solKK]
```

$$\frac{\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{2\,A^{2}\,b\,(-1+\alpha)}$$

利潤最大をもたらす労働投入量，LLopt，は L=K/(K/L) より求めることができる。

```
LLopt=Simplify[KKopt/KoverL]
```

$$-\frac{r\alpha\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{2\,A^{2}\,b\,w\,(-1+\alpha)^{2}}$$

```
LLopt2=Simplify[PowerExpand[KKopt/KoverL]]
```

$$-\frac{r^{1-2\alpha}\alpha^{1-2\alpha}(w-w\alpha)^{\alpha}\left(a\,A\,r^{\alpha}(-1+\alpha)\,\alpha^{\alpha}+r\,(w-w\alpha)^{\alpha}\right)}{2\,A^{2}\,b\,w\,(-1+\alpha)^{2}}$$

利潤最大をもたらす生産量，YYopt，は生産関数 Y[L,K] より

```
YYopt=Simplify[Y[LLopt,KKopt]]
```

$$\frac{1}{2}A\left(\frac{\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{1-\alpha}\left(-\frac{r\alpha\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}$$

```
YYopt2=Simplify[PowerExpand[Y[LLopt,KKopt]]]
```

$$\frac{(-1)^{\alpha}r^{-\alpha}w^{-\alpha}(-1+\alpha)^{-1-\alpha}\alpha^{-\alpha}(w-w\alpha)^{\alpha}\left(a\,A\,r^{\alpha}(-1+\alpha)\,\alpha^{\alpha}+r\,(w-w\alpha)^{\alpha}\right)}{2\,A\,b}$$

さらに，利潤最大をもたらす価格，priceopt，は逆需要関数に生産量の値を代入することで求められる。

```
priceopt=a-b YYopt
```

$$a-\frac{1}{2}A\,b\left(\frac{\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{1-\alpha}$$

$$\left(-\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}$$

priceopt2=Simplify[PowerExpand[priceopt]]

$$\frac{1}{2}\left(2\,a-\frac{(-1)^{\alpha}\,r^{-\alpha}\,w^{-\alpha}\,(-1+\alpha)^{-1-\alpha}\,\alpha^{-\alpha}\,(w-w\,\alpha)^{\alpha}\,\left(a\,A\,r^{\alpha}\,(-1+\alpha)\,\alpha^{\alpha}+r\,(w-w\,\alpha)^{\alpha}\right)}{A}\right)$$

また，最大利潤，paiopt，は労働投入量，資本投入量，および生産量の値を利潤の定義式に代入することで求められる。

paiopt=Simplify[(a-b YYopt)YYopt -(w LLopt+r KKopt)]

$$\frac{1}{2\,A^{2}\,b\,(-1+\alpha)^{2}}\left(\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)\right.$$
$$\left(r+A\,(-1+\alpha)\left(\frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{-\alpha}\right.$$
$$\left(-\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}$$
$$\left(a-\frac{1}{2}\,A\,b\left(\frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{1-\alpha}\right.$$
$$\left.\left.\left.\left(-\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}\right)\right)\right)$$

paiopt2=Simplify[PowerExpand[(a-b YYopt)YYopt -(w LLopt+r KKopt)]]

$$-\frac{1}{4\,A^{2}\,b}\Big(r^{-2\,\alpha}\,w^{-2\,\alpha}\,(-1+\alpha)^{-2\,(1+\alpha)}\,\alpha^{-2\,\alpha}\,(w-w\,\alpha)^{\alpha}\,\left(a\,A\,r^{\alpha}\,(-1+\alpha)\,\alpha^{\alpha}+r\,(w-w\,\alpha)^{\alpha}\right)$$
$$\Big((-1)^{\alpha}\,a\,A\,r^{\alpha}\,(-1+\alpha)\,\alpha^{\alpha}\,\left(-2\,w^{\alpha}\,(-1+\alpha)^{\alpha}+(-1)^{\alpha}\,(w-w\,\alpha)^{\alpha}\right)+$$
$$r\,\left(-2\,w^{2\,\alpha}\,(-1+\alpha)^{2\,\alpha}+(-1)^{2\,\alpha}\,(w-w\,\alpha)^{2\,\alpha}\right)\Big)\Big)$$

利潤を最大にする労働投入量LLfun，資本投入量KKfun，生産量YYfun，価格pricefun，および利潤paifunをパラメータ A，α，a，b，w，およびrであらわそう。このようにしておくと，任意のパラメータの組に対応した，利潤最大をもたらす労働投入量，資本投入量，生産量，価格，およびそのとき

の利潤水準を求めるときに便利である。

$$\mathbf{LLfun[A\_,\ \alpha\_,\ a\_,\ b\_,\ w\_,\ r\_]} := -\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{2\,A^{2}\,b\,w\,(-1+\alpha)^{2}}$$

$$\mathbf{KKfun[A\_,\ \alpha\_,\ a\_,\ b\_,\ w\_,\ r\_]} := \frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{2\,A^{2}\,b\,(-1+\alpha)}$$

$$\mathbf{YYfun[A\_,\ \alpha\_,\ a\_,\ b\_,\ w\_,\ r\_]} :=$$

$$\frac{1}{2}\,A\left(\frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{1-\alpha}$$
$$\left(-\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}$$

$$\mathbf{pricefun[A\_,\ \alpha\_,\ a\_,\ b\_,\ w\_,\ r\_]} :=$$

$$a-\frac{1}{2}\,A\,b\left(\frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{1-\alpha}$$
$$\left(-\frac{r\,\alpha\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,w\,(-1+\alpha)^{2}}\right)^{\alpha}$$

$$\mathbf{paifun[A\_,\ \alpha\_,\ a\_,\ b\_,\ w\_,\ r\_]} :=$$

$$\frac{1}{2\,A^{2}\,b\,(-1+\alpha)^{2}}$$
$$\left(\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)\right.$$
$$\left(r+A\,(-1+\alpha)\left(\frac{\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{-2\,\alpha}\left(r+a\,A\,(-1+\alpha)\left(\frac{r\,\alpha}{w-w\,\alpha}\right)^{\alpha}\right)}{A^{2}\,b\,(-1+\alpha)}\right)^{-\alpha}\right.$$

$$\left(-\frac{r\alpha\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+aA(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}bw(-1+\alpha)^{2}}\right)^{\alpha}$$

$$\Bigg(a-\frac{1}{2}Ab\Bigg(\left(\frac{\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+aA(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}b(-1+\alpha)}\right)^{1-\alpha}$$

$$\left(-\frac{r\alpha\left(\frac{r\alpha}{w-w\alpha}\right)^{-2\alpha}\left(r+aA(-1+\alpha)\left(\frac{r\alpha}{w-w\alpha}\right)^{\alpha}\right)}{A^{2}bw(-1+\alpha)^{2}}\right)^{\alpha}\Bigg)\Bigg)\Bigg)\Bigg)$$

A＝1，α＝0.6，a＝50，b＝1，w＝200，r＝0.1のとき，独占企業の利潤を最大にする労働投入量，資本投入量，生産量，価格，および利潤は次のようにMATHEMATICAに入力すると得られる[21)]。

```
A=1;α=0.6;(* production function Y=AL^αK^(1-α) *)
a=50;b=1;(* demand curve p=a - bY *)
w=200;r=0.1;(* cost CC=wL+rK *);

   LLoptNum = LLfun[1,0.6,50,1,200,0.1]
   0.878827
   KKoptNum = KKfun[1,0.6,50,1,200,0.1]
   1171.77
   YYoptNum = YYfun[1,0.6,50,1,200,0.1]
   15.6271
   priceoptNum = pricefun[1,0.6,50,1,200,0.1]
   34.3729
   paioptNum = paifun[1,0.6,50,1,200,0.1]
   244.207
```

まとめて示すと，

21）MATHEMATICAでは，(*　*）の部分はコメント行なので，結果を求めるためには入力の必要はない。

```
{LLoptNum,KKoptNum,YYoptNum,priceoptNum,paioptNum}
{{0.878827},{1171.77},{15.6271},{34.3729},{244.207}}
```

となる。

以上の結果より，収入を最大にする場合と利潤を最大にする場合をここでまとめておく。表1より明らかなように，コブ＝ダグラス生産関数の場合は，収入を最大にする生産量は一意に決まるが，その生産量を生産する方法（労働投入量と資本投入量の組（L,K））は無限にある。したがって，資本－労働比率は一意に決まらない。他方，利潤を最大にする生産量は，資本－労働比率，労働投入量，資本投入量とともに一意に決まる。

生産量は収入を最大にする場合のほうが利潤を最大にする場合よりも多い。したがって，価格は収入を最大にする場合のほうが利潤を最大にする場合より安くなる。

表1　収入最大化と利潤最大化の比較

	収入最大化	利潤最大化
資本－労働比率	一意に決まらない	1333.333
労働投入量	一意に決まらない	0.879
資本投入量	一意に決まらない	1171.769
生産量	25	15.627
価格	25	34.373
収入の最大値	625	
利潤の最大値		244.207

3．収入曲面と（費用平面に平行な）補助平面を利用して，最大利潤を探す

この節では，利潤は収入から費用を減じたものであることを考慮し，最大利潤をもたらす労働投入量と資本投入量の組（L,K）を求める。

MATHEMATICA で収入曲面を描かせ，費用平面に平行な補助平面[22)]で収入曲面を切断する。さらに，補助平面の水準を次々に変化させたアニメーションを用いて，最大利潤をもたらす労働投入量と資本投入量の組（L,K）を求める。

図55から図63までからわかるように，補助平面の水準を上昇させると，その切断面より上に見える収入曲面の一部分はだんだん小さくなっていることに注意しよう。したがって，補助平面による切断面の上に収入曲面の一部が見えているときは，利潤最大になっていないこともわかる。

MATHEMATICA で作成した正しい画像では，利潤最大点は補助平面に接する状態になっているので見ることはできないが，図64に，参考のために利潤最大点の位置を示した。



図55　利潤＝0



図56　利潤＝100



図57　利潤＝200

22）画像では黒色の平面である。



図58　利潤＝220



図59　利潤＝240



図60　利潤＝241



図61　利潤＝242



図62　利潤＝243



図63　利潤＝244.2



図64　利潤＝244.3

４．利潤曲面と（利潤水準に対応する）補助平面を利用して，最大利潤を探す

この節では，MATHEMATICA で利潤曲面を直接描かせ，ある利潤水準に対応する補助平面で利潤曲面を切断する。さらに，この利潤水準を次々に変化させたアニメーションを用いて，最大利潤をもたらす労働投入量と資本投入量の組（L,K）を求める。

図65と図66に利潤曲面の一部分と利潤最大点の近傍を示した。

図67より図71において，（利潤水準を示す透明な補助平面の）切断でできた切り口が見えているときは，利潤最大になっていないことに注意しよう。図72



図65　公正報酬率規制のないときの利潤曲面の一部



図66　公正報酬率規制のないときの利潤曲面（利潤最大点の近傍）

は利潤最大点が達成されていることを示している。

同様に，図73より図77において，（利潤水準を示す青色の補助平面による）切断面の上に利潤曲面の一部が見えているときは，利潤最大になっていないことがわかる。図78は利潤最大点が達成されていることを示している。



図67　利潤＝228.207



図68　利潤＝236.207



図69　利潤＝241.207



図70　利潤＝242.207



図71　利潤＝243.207



図72　利潤＝244.207　利潤最大点（青丸印）が達成される



図73　利潤＝156.707



図74　利潤＝226.707



図75　利潤＝236.707



図76　利潤＝239.707



図77　利潤＝239.207



図78　利潤＝244.207　利潤最大点（赤丸印）が達成されている

5．制約の効果 ―― アバーチ＝ジョンソン・モデル

5．1　公正報酬率規制とは

Averch and Johnson［1962］は公正報酬率規制の下での利潤最大化を検討し，規制を加えた場合のほうが規制のない場合に比べて，資本投入量が多くなることを明らかにした[23)]。

公正報酬率規制とは，規制当局がある独占企業に対して，ある公正報酬率 s を設定し，次の条件をその企業が満たすよう求めているときをいう。その条件とは，収入から賃金支払いをした残額は，現存する資本額に公正報酬率 s を乗じた値を超えてはいけない。すなわち，

$$P(F(L,K))F(L,K)-wL \leqq sK \qquad \text{（一般の場合）}$$

$$(50-Y)Y-200L \leqq 0.2K \qquad \text{（数値例の場合）} \qquad (1)$$

という規制である。ここに，P(F(L,K)) は逆需要関数[24)]，F(L,K) は生産量，

23) Averch and Johnson［1962］の論証の不備を指摘した Takayama［1969］は，Kuhn-Tucker-Lagrange 条件を用いて，厳密な論証をしている。なお，Zajac［1970］は図形的にこの問題を検討している。

24) ここでの数値例の場合は p＝50－Y である。ただし，p は価格，Y は生産量を示す。

Ｌは労働投入量，Ｋは資本投入量，およびｗは賃金率を示す。

独占企業の目的は規制の下で利潤 π，すなわち，

$\pi = P(F(L,K))F(L,K) - wL - rK$，ただし，ｒはレンタル料である

（一般の場合）

$\pi = (50 - Y)Y - 200L - 0.1K$，ただし，$Y = L^{0.6}K^{0.4}$である

（数値例の場合）　(2)

を最大化することである。

数値例について，Kuhn-Tucker-Lagrange 条件を用いて上の問題を解こう[25)]。Lagrange 乗数を λ とすると，労働投入量に関する条件より，

$$(\lambda - 1)[(50 - 2Y)\partial Y/\partial L - 200] = 0 \tag{3}$$

また，資本投入量に関する条件より，

$$[(50 - 2Y)\partial Y/\partial K - 0.1] - \lambda[(50 - 2Y)\ \partial Y/\partial K - 0.2] \leqq 0 \tag{4}$$

λ に関する条件より，

$$(50 - Y)Y - 200L \leqq 0.2K \tag{5}$$

を得る。

なお，生産関数より，

$$Y = L^{0.6}K^{0.4} \tag{6}$$

(4)より，$\lambda \neq 1$だから，(3)より，

$$(50 - 2Y)\partial Y/\partial L = 200 \tag{7}$$

(6)より，$\partial Y/\partial L = 0.6(K/L)^{0.4}$　(8)

25）Takayama［1969］を参照せよ。以下では厳密さを捨象し，必要な式の展開のみを述べる。

(8)を(7)に代入すると，Y を(K/L)であらわすことができる。

$$Y = 25 - 500/3(K/L)^{-0.4} \tag{9}$$

(5)は等式で成立するので[26)]，両辺を L で割り，移行すると，

$$(50 - Y)(Y/L) - 200 - 0.2(K/L) = 0 \tag{10}$$

(6)より，$(Y/L) = (K/L)^{0.4}$となるので，(9)を(10)に代入し，整頓すると，

$$0.2(K/L) - 25(K/L)^{0.4} + 100/3 = 0 \tag{11}$$

(11)より資本－労働比率 K/L が求まり，(9)に代入すると生産量 Y が求まる。$Y = L(K/L)^{0.4}$および $Y = K(K/L)^{-0.6}$より，それぞれ労働投入量 L と資本投入量 K が決まる。これらの値を逆需要関数，price＝50－Y，および利潤の定義式(2)に代入すると，価格 price と利潤 profit が求まる。

MATHEMATICA を利用して，最初に(11)を満たす，**k**≡K/L を求めよう。

```
f[k_]:=0.2 k - 25 k^(2/5)+100/3
sol=Solve[f[k]==0,k]
{{k→2.11866},{k→2841.83}}
k=k/.sol
{2.11866,2841.83}
Y=25-(500/3) k^(-0.4)
{-98.4309,18.0752}
```

(11)を満たす **k** は 2 つあるが，そのうち小さい値は生産量 **Y** を負にするので除外する。

したがって，

26) Takayama［1969］を参照せよ。

```
k=k[[2]]
2841.83
Y=Y[[2]]
18.0752
L=Y/k^0.4
0.751006
K=Y k^0.6
2134.23
price=50-Y
31.9248
profit=(50-Y) Y -(200 L + 0.1 K)
213.423
```

となる。

5．2　アニメーションで公正報酬率規制の下での利潤最大化を求める

5．1の結果を MATHEMATICA で作成したアニメーション画像で確認しよう。

図79より図82において，公正報酬率規制は斜めの「赤紫のグラデーション」平面で示され，利潤曲面は黄緑色，利潤水準を示す補助平面は濃い青色で示さ



図79　利潤＝117.423



図80　利潤＝197.423



図81　利潤＝209.423



図82　利潤＝213.423（公正報酬率規制のあるときの最大利潤）

れている。利潤水準が増加するにつれて，黄色の丸印（○）で示される最大利潤点[27)]だけが残っていくことが読み取れる。

図83より図85において，（利潤水準を示す透明な補助平面の）切断でできた切り口の中側を（公正報酬率規制を示す）制約平面が通っているので，利潤最大になっていないことがわかる。図86に示したように，利潤最大点では（公正報酬率規制を示す）制約平面が利潤曲面を切る縁のところを（利潤水準を示す）透明な補助平面が切断している。わかりやすくするために，利潤最大点を示す（黄色の）丸印（○）を示してある。



図83　利潤＝208.423



図84　利潤＝209.423

27）公正報酬率規制を示す斜めの「赤紫のグラデーション」平面と利潤曲面を示す黄緑色の曲面の，境界部分である。



図85　利潤＝211.423



図86　利潤＝213.423（公正報酬率規制のあるときの最大利潤）

5．3　公正報酬率規制のもとで利潤最大化をもたらす資本投入量を求める

資本投入量の水準を示す，L－K 平面に垂直な補助平面で利潤曲面を切断する。この補助平面を移動することで，利潤最大点を含む補助平面を見つける。図87から図92まではこのプロセスを表示したもので，公正報酬率規制の下で利潤最大化をもたらす資本投入量を見つけることができる。

公正報酬率規制のあるときの利潤最大点をもたらす資本量を表示している図92は図91を左奥より見たものである。



図87　資本投入量＝1334.23



図89　資本投入量＝1894.23



図88　資本投入量＝1734.2



図90　資本投入量＝1974.23



図91　資本投入量＝2134.23（公正報酬率規制のあるときの利潤最大点をもたらす資本投入量）



図92　資本投入量＝2134.23（公正報酬率規制のあるときの利潤最大点をもたらす資本投入量）

5．4　公正報酬率規制のもとで利潤最大化をもたらす労働投入量を求める

労働投入量の水準を示す，L－K 平面に垂直な補助平面で利潤曲面を切断する。この補助平面を移動することで，利潤最大点を含む補助平面を見つける。図93から図96まではこのプロセスを表示したもので，公正報酬率規制の下で利潤最大化をもたらす労働投入量を見つけることができる。



図93　労働投入量＝1.95101



図94　労働投入量＝1.40101



図95　労働投入量＝0.901005



図96　労働投入量＝0.751005（公正報酬率規制のあるときの利潤最大点をもたらす労働投入量）

5．5　公正報酬率規制のない場合と公正報酬率規制のある場合の均衡値を比較

公正報酬率規制のない場合と公正報酬率規制のある場合の均衡値[28)]を比較するために，第2節から第5節までの結果を表2のようにまとめた。

Averch and Johnson［1962］が主張するように，公正報酬率規制のある場合は規制のない場合に比べて，資本投入量が大きくなっている。また，生産量も規制のあるほうが規制のない場合に比して増加している。

図97および図98を注意深く見ると，労働投入量は規制のある場合のほうが規制のない場合よりも少ない。資本投入量は明らかに，規制のある場合のほうが

28）公正報酬率規制のない場合と公正報酬率規制のある場合において，利潤最大をもたらす，労働投入量，資本投入量，生産量，価格および利潤の組をそれぞれの均衡値とする。資本－労働比率は参考のために掲示した。

表２　公正報酬率規制のない場合と公正報酬率規制のある場合の均衡値の比較

	規制のない場合	大小関係	規制のある場合
資本－労働比率（K/L）	1333.333	＜	2841.820
労働投入量（L）	0.879	＞	0.751
資本投入量（K）	1171.769	＜	2134.224
生産量（Y）	15.627	＜	18.075
価格（＝50－生産量）	34.373	＞	31.925
利潤	244.207	＞	213.423

規制のない場合よりも多くなっている。また，図97において，利潤水準を示す平面の上に，ゴルフ場のグリーンのような形が見えていることから，利潤は規制のある場合のほうが規制のない場合よりも少なくなっている。さらに，図98において，規制のある場合の均衡点は黒丸印（●）で示され，規制のない場合の均衡点は白丸印（○）で示されている。したがって，等生産量曲線の位置からわかるように，生産量は規制のある場合のほうが規制のない場合よりも多くなっている。需要曲線は右下がりであることを考慮すると，価格は規制のある場合のほうが規制のない場合よりも低くなっている。原点と均衡点を結ぶ直線の傾きは資本－労働比率を示すので，資本－労働比率（K/L）は規制のある場合のほうが規制のない場合よりも大きくなっている。



図97　規制のない場合とある場合の均衡点の比較



図98　規制のない場合とある場合の投入量の組み合わせ，および生産量の比較

結　　　語

この小論では，生産理論のいくつかの話題をMATHEMATICAの数式処理とアニメーション画像で示すことを試みてみた。

CES生産関数の3次元空間におけるグラフ表現は初めての試みである。特に，代替の弾力性に関するパラメータ ρ と分配にかかわるパラメータ α の値の変化とともに，その形状が移り行く様を（部分的であるが）アニメーションで示した。数式展開では明らかだった内容も，アニメーション画像で確認することで理解度を増すことができる。

なお，グラフだけでは微妙な数値の違いを読み取ることは一般には難しい。均衡の比較，すなわち，比較静学の場合は，MATHEMATICAによる数式処理が大いに活躍する。

また，カラーで作成した画像を白黒印刷するとき，カラー作成時に予想した「白黒コントラスト」の画像が意外と鮮明な印刷とならない場合が多い。この点は，インターネット上にアニメーション画像を提供することで補完したいと思う。

参考文献

Arrow, K. J., H. B. Chenery, B. S. Minhas, and R. M. Solow [1961], "Capital-labor substitution and economic efficiency," *Review of Economics and Statistics*, Vol.43, No.3, pp.225-250. この論文は *Collected Papers of Kenneth J. Arrow*, Volume 5 Production and Capital, Chapter 3, pp.50-103, The Belknap Press of Harvard University Press, 1985 に再録されている。

Averch, Harvey and Leland L. Johnson [1962], "Behavior of the Firm Under Regulatory Constraint," *American Economic Review*, Vol.52, pp.1052-1069.

Cobb, C. W. and P. H. Douglas [1928], "A Theory of Production," *American Economic Review, Supplement*, Vol.18, No.1, pp.139-165.

Christensen, L. R., D. W. Jorgenson, and L. J. Lau [1971], "Conjugate Duality and the Transcendental Logarithmic Production Function," *Econometrica*, Vol. 39, No.4 (July), pp.225-256.

Christensen, L. R., D. W. Jorgenson, and L. J. Lau [1973], "Transcendental Logarithmic Production Frontiers," *Review of Economics and Statistics*, Vol.55, No.1 (February), pp.28-45.

Diewert, W. E. [1982], "Duality Approaches to Microeconomic Theory," in *Handbook of Mathematical Economics*, Vol. II, edited by K. J. Arrow and M. D. Intriligator, Chapter 12, pp.535-599, North-Holland Publishing Company, Amsterdam.

Huang, Cliff J. and Philip S. Crooke [1997], *Mathematics and Mathematica for Economists*, Blackwell Publishers Ltd.

Nadiri, M. Ishaq [1982], "Producers Theory," in *Handbook of Mathematical Economics*, Vol. II, edited by K. J. Arrow and M. D. Intriligator, Chapter 10, pp.431-490, North-Holland Publishing Company, Amsterdam.

Sato, K. [1975], *Production Functions and Aggregation*, North-Holland Publishing Company, Amsterdam.

Shephard, R. W. [1953], *Cost and Production Functions*, Princeton: Princeton University Press.

Shephard, R. W. [1970], *Theory of Cost and Production Functions*, Princeton: Princeton University Press.

Stinespring, John Robert [2002], *MATHEMATICA for MICROECONOMICS Learning by Example*, Academic Press, A division of Harcourt, Inc., San Diego, USA.

Takayama, Akira [1969], "Behavior of the Firm Under Regulatory Constraint," *American Economic Review*, Vol.59, pp.255-260.

Varian, Hal R. ed. (1993), *Economic and Financial Modeling with Mathematica*, Santa Clara, CA, TELOS, Springer-Verlag New York, Inc.

Varian, Hal R. ed. (1996), *Computational Economics and Finance Modeling and Analysis with Mathematica*, Santa Clara, CA, TELOS, Springer-Verlag New York, Inc.

Wickham-Jones, Tom [1994], *Mathematica Graphics Technique & Applications*, Santa Clara, CA, TELOS, Springer-Verlag New York, Inc.

Wolfram, Stephen [1999], *The MATHEMATICA BOOK*, Fourth Edition, Wolfram Media and Cambridge University Press.

Zajac, E. E. [1970], "A Geometric Treatment of Averch-Johnson's Behavior of the Firm Model," *American Economic Review*, Vol.60, pp.117-125.

浅利一郎・久保徳次郎・石橋太郎・山下隆之［1997］『はじめよう　経済学のための Mathematica　パソコンによる数式処理』，日本評論社。

鵜沢秀［1996］「Mathematica を用いてクールノー均衡とシュタッケルベルク均衡を求める」1994−1995年度特定研究報告書『数式処理言語 Mathematica を利用した数理モデルの解析』（小樽商科大学），pp.35−55.

鵜沢秀［1998］「数式処理システム Mathematica の応用とインターネットを利用した経済学学習について」，『商学討究』（小樽商科大学），第48巻第２，３合併号，pp.47−72.

鵜沢秀［2000a］「MATHEMATICA のグラフィックス機能を利用してクールノー均衡を見る —— あなたがクールノー・モデルのパラメータを変更できる ——」，『情報処理センター　広報』（小樽商科大学）第13号，pp.11−25.

鵜沢秀［2000b］『産業組織論』（エコノミスト社）

鵜沢秀［2002］「シュタッケルベルク均衡をアニメーションで見る —— MATHEMATICA，HTML，および JavaScript を用いて ——」『商学討究』（小樽商科大学），第53巻第１号，pp.33−75.

鵜沢秀［2005］「MATHEMATICA と LiveGraphics3D の応用 —— クールノー均衡を例にして ——」『商学討究』（小樽商科大学），第55巻第４号，pp.85−106.

鵜沢秀［2006］「MATHEMATICA の経済学への応用⑴ —— コブ＝ダグラス型効用関数の場合の消費者均衡理論 ——」『商学討究』（小樽商科大学），第56巻第４号，pp.37−74.

鵜沢秀［2007］「MATHEMATICA の経済学への応用⑵ —— ギッフェン財をもたらす効用関数の特徴について ——」『商学討究』（小樽商科大学），第57巻第４号，pp.63−101.

小林道正［1996］『Mathematica による『ミクロ経済学』スタディガイド』（東洋経済新報社）

西村和雄［1990］『ミクロ経済学』（東洋経済新報社）

早見弘，鵜沢秀，若林信夫，今喜典，佐竹正夫［1986］『現代経済学講義』（中央経済社）